# NAG Fortran Library Routine Document

# F08GNF (ZHPEV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F08GNF (ZHPEV) computes all the eigenvalues and, optionally, eigenvectors of a complex $n$ by $n$ Hermitian matrix in packed storage.

## 2 Specification

```
SUBROUTINE F08GNF (JOBZ, UPLO, N, AP, W, Z, LDZ, WORK, RWORK, INFO)
INTEGER            N, LDZ, INFO
double precision   W(*), RWORK(*)
complex*16         AP(*), Z(LDZ,*), WORK(*)
CHARACTER*1        JOBZ, UPLO
```

The routine may be called by its LAPACK name ***zhpev***.

## 3 Description

The Hermitian matrix $A$ is first reduced to real tridiagonal form, using unitary similarity transformations, and then the $QR$ algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

1: JOBZ – CHARACTER*1 *Input*

*On entry*: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

*Constraint*: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER*1 *Input*

*On entry*: if UPLO = 'U', the upper triangle of $A$ is stored.

If UPLO = 'L', the lower triangle of $A$ is stored.

3: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: N $\geq$ 0.

4: AP($*$) – ***complex*16*** array *Input/Output*

**Note**: the dimension of the array AP must be at least $\max(\mathrm{N} \times (\mathrm{N}+1)/2)$.

*On entry*: the upper or lower triangle of the Hermitian matrix $A$, packed columnwise in a linear array. The $j$th column of $A$ is stored in the array AP as follows:

if UPLO = 'U', $\mathrm{AP}(i + (j-1) \times j/2) = a_{ij}$ for $1 \leq i \leq j$;
if UPLO = 'L', $\mathrm{AP}(i + (j-1) \times (2 \times n - j)/2) = a_{ij}$ for $j \leq i \leq n$.

*On exit*: is overwritten by values generated during the reduction to tridiagonal form. If UPLO = 'U' , the diagonal and first super-diagonal of the tridiagonal matrix $T$ overwrite the corresponding elements of $A$, and if UPLO = 'L', the diagonal and first sub-diagonal of $T$ overwrite the corresponding elements of $A$.

5: W($*$) – ***double precision*** array *Output*

**Note**: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

*On exit*: if INFO = 0, the eigenvalues in ascending order.

6: Z(LDZ,$*$) – ***complex*16*** array *Output*

**Note**: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

*On exit*: if JOBZ = 'V', then if INFO = 0, Z contains the orthonormal eigenvectors of the matrix $A$, with the $i$th column of $Z$ holding the eigenvector associated with $\mathrm{W}(i)$.

If JOBZ = 'N', Z is not referenced.

7: LDZ – INTEGER *Input*

*On entry*: the first dimension of the array Z as declared in the (sub)program from which F08GNF (ZHPEV) is called.

*Constraints*:

if JOBZ = 'V', LDZ $\geq \max(1, \mathrm{N})$;
LDZ $\geq 1$ otherwise.

8: WORK($*$) – ***complex*16*** array *Workspace*

**Note**: the dimension of the array WORK must be at least $\max(1, 2 \times \mathrm{N} - 1)$.

9: RWORK($*$) – ***double precision*** array *Workspace*

**Note**: the dimension of the array RWORK must be at least $\max(1, 3 \times \mathrm{N} - 2)$.

10: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th argument had an illegal value.

INFO $> 0$

If INFO $= i$, the algorithm failed to converge; $i$ off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A+E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and $\epsilon$ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

Each eigenvector is normalized so that the element of largest absolute value is real and positive.

The total number of floating point operations is proportional to $n^3$.

The real analogue of this routine is F08GAF (DSPEV).

## 9 Example

To find all the eigenvalues of the Hermitian matrix

$$A = \begin{pmatrix} 1 & 2-i & 3-i & 4-i \\ 2+i & 2 & 3-2i & 4-2i \\ 3+i & 3+2i & 3 & 4-3i \\ 4+i & 4+2i & 4+3i & 4 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08GNF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=10)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      DOUBLE PRECISION EERRBD, EPS
      INTEGER          I, INFO, J, N
*     .. Local Arrays ..
      COMPLEX *16      AP((NMAX*(NMAX+1))/2), DUMMY(1,1), WORK(2*NMAX-1)
      DOUBLE PRECISION RWORK(3*NMAX-2), W(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         ZHPEV
*     .. Intrinsic Functions ..
      INTRINSIC        ABS, MAX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08GNF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the upper or lower triangular part of the matrix A from
*        data file
*
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+(J*(J-1))/2),J=I,N),I=1,N)
```

```
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+((2*N-J)*(J-1))/2),J=1,I),I=1,N)
         END IF
*
*        Solve the Hermitian eigenvalue problem
*
         CALL ZHPEV('No vectors',UPLO,N,AP,W,DUMMY,1,WORK,RWORK,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (W(J),J=1,N)
*
*           Get the machine precision, EPS and compute the approximate
*           error bound for the computed eigenvalues.  Note that for
*           the 2-norm, max( abs(W(i)) ) = norm(A), and since the
*           eigenvalues are returned in ascending order
*           max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)))
*
            EPS = X02AJF()
            EERRBD = EPS*MAX(ABS(W(1)),ABS(W(N)))
*
*           Print the approximate error bound for the eigenvalues
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Error estimate for the eigenvalues'
            WRITE (NOUT,99998) EERRBD
         ELSE
            WRITE (NOUT,99997) 'Failure in ZHPEV. INFO =', INFO
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
      END
```

## 9.2 Program Data

```
F08GNF Example Program Data

  4                                                 :Value of N

 (1.0, 0.0)  (2.0, -1.0)  (3.0, -1.0)  (4.0, -1.0)
             (2.0,  0.0)  (3.0, -2.0)  (4.0, -2.0)
                          (3.0,  0.0)  (4.0, -3.0)
                                       (4.0,  0.0) :End of matrix A
```

## 9.3 Program Results

```
 F08GNF Example Program Results

 Eigenvalues
    -4.2443 -0.6886  1.1412 13.7916

 Error estimate for the eigenvalues
        1.5E-15
```